

**1. Konferenz der SAS-Benutzer in Forschung und Entwicklung (KSFE), Humboldt-Universität zu Berlin, 20./21. Februar 1997**

**Videobildanalyse zur Bestimmung des Marmorierungsgrades im Kotelett mit Hilfe verschiedener SAS-Prozeduren**

Armin Scholz

Humboldt-Universität zu Berlin, Institut für Nutztierwissenschaften,  
 FG Populationsgenetik, Lentzeallee 75, 14195 Berlin \*  
 email: schohub1@mailszrz.zrz.tu-berlin.de

**Einleitung**

Ein wichtiges Kriterium für den Genußwert des Schweinefleisches, den Geschmack und die Beschaffenheit stellt das intramuskuläre Fett bzw. die Marmorierung im Kotelettmuskel (Musculus longissimus dorsi) dar. In der praktischen Schweinezucht wurde aufgrund der Preisgestaltung für Schlachtschweine der Hauptselektionserfolg im Magerfleischanteil erzielt. Infolgedessen weist der intramuskuläre Fettgehalt (IMF) im Kotelettmuskel der in Deutschland hauptsächlich zur Fleischerzeugung eingesetzten Schweinerassen bzw. Kreuzungen bei einer Variation von durchschnittlich 0,6 - 1,8 % ein sehr niedriges Niveau auf. Die einzige Ausnahme bildet die Rasse Duroc mit einem IMF von ca. 2,5%. Das Optimum für den intramuskulären Fettgehalt wird jedoch bei 1,5 - 2,5 % angegeben.

Ein weiteres Problem liegt in der ungenauen Bestimmung des intramuskulären Fettgehaltes bzw. des Marmorierungsgrades. Gegenwärtig wird in den deutschen Schweine-Leistungsprüfanstalten nur eine subjektive Bonitierung der Marmorierung (Noten 1 - 6) am Kotelettanschnitt zwischen 13. und 14. Brustwirbel vorgenommen.

Eine Umkehr bei der Entwicklung des intramuskulären Fettgehaltes ist erst durch Einführung einer objektiven Leistungsprüfung und die Einbeziehung dieses Merkmals in die Selektion zu erwarten. Zur Bestimmung des intramuskulären Fettgehaltes am Schlachtkörper bzw. am lebenden Tier werden verschiedene chemische und physikalische Methoden genutzt, wie z. B. die Soxhlet - Extraktion, Ultraschall - Technik, Reflexionsmessungen, Computer-Tomographie, 1H-NMR- bzw. Infrarot - Spektroskopie.


Da in einigen Prüfstationen die Videobildanalyse (VBA) bereits zur Ermittlung der Fett- und Fleischfläche am Kotelettanschnitt eingesetzt wird, war es naheliegend, diese Technik ebenfalls für die Ermittlung des Marmorierungsgrades zu nutzen.

**Material und Methode**

Für die Untersuchung wurden 130 Koteletts (13./14. Brustwirbel) von Schweinen verschiedener Genotypen aus der Leistungsprüfanstalt Ruhlsdorf genutzt. Die Koteletts wurden mit einer 3 Chip R G B - Color - Kamera (Hitachi - HV-C10) beidseitig cranial bzw. caudal aufgenommen und über drei Analog - Digital - Wandler (ITI - Color Frame Grabber) auf einer wiederbeschreibbaren Magneto-Optischen-Diskette (600 MB) gespeichert. Ein Bild umfaßt 768\*512 Bildpunkte bei einer Datentiefe von 24 bit, die auf die drei Farbkanäle **ROT**, **GRÜN** und **BLAU** (R G B) aufgeteilt sind. Für jeden Bildpunkt, der eine Fläche von ca. 0,05 mm<sup>2</sup> einnimmt, liegen folglich drei Grauwertinformationen jeweils im Bereich von 0 - 255 vor.

## Videobild eines Koteletts mit 3-Chip-RGB-Kamera

(768\*512 Pixel)




Mit Hilfe des Bildverarbeitungsprogrammes BioScan-OPTIMAS wurden die interessierenden Kotelettregionen, sogenannte Regions Of Interest (ROI's), halbautomatisch eingegrenzt, maskiert und als neue TIF-Datei gespeichert.

## Maskieren (Ausschneiden) der Region of Interest (ROI)



Die Kantendetektion beruht dabei auf den Grauwertunterschieden zwischen Fleisch- und Fettgewebe. Niedrige (dunkle) Grauwerte charakterisieren Fleischgewebe und hohe Grauwerte kennzeichnen Fettgewebe.

## Ergebnis der Maskierung



Die Weiterverarbeitung der TIF-Bilddatei erfolgte mit SAS 6.11. Anhand einer Beispiel-TIF-Datei (f26lmsk.tif) wird der Ablauf nachfolgend näher beschrieben.

### **/\* 1. Schritt: Bildinformationen filtern - Bildaufbau in Tags, Stripes und Scores kodiert \*/**

```
%let file=d:\mp120193\f26lmsk.tif;

libname f26lmsk "d:\asda";
filename tif "&file";
title "&file";
data tags (keep=tag text type typ length score comment)
  stripes (keep=stripa stripe stripl)
  scores (keep=subfile xpixels ypixels bits compress photo nrows
          minval maxval xres yres planar resunit nstrips);
  length text comment modulus $32;
  length default=4;
  retain ifd ntags (2 0);
  infile tif recfm=n; /* ermöglicht das Importieren von beliebigen internen Binär - Daten */
                    /* header */
  input fa $char2. vers pib2.;
  if fa='II' then firma='Intel ';
  if fa='MM' then firma='Motorola';

a: iifd+1;
  iifd+1;
  input @(ifd+12*ntags+3) ifd pib4.;
  if ifd=0 then stop;
                    /* ifd */
  input @(ifd+1) ntags pib2.;
do itag=1 to ntags;
  comment="";
  input @(ifd+12*itag-9) tag pib2. type pib2. length pib4. score pib4.;
  select (tag);
```

```

when (254) text='NewSubfile';
when (255) do;
    text='SubfileType';
    select (score);
        when (1) modus='FullResolution';
        when (2) modus='ReducedResolution';
        when (3) modus='SinglePageImage';
        otherwise modus='unknown';
    end;
    comment=modus;
    subfile=score;
end;
when (256) do;          /* Breite */
    text='ImageWidth';
    comment='Xpixels';
    xpixels=score;
end;
when (257) do;          /* Höhe */
    text='ImageLength';
    comment='Ypixels';
    ypixels=score;
end;
when (258) do;
    text='BitsPerSample';
    bits=score;
end;
when (259) do;
    text='Compression';
    select (score);
        when (1) modus='Uncompressed';
        when (2) modus='Huffman RLE';
        when (3) modus='FAX CCITT Group3';
        when (4) modus='FAX CCITT Group4';
        when (5) modus='LZW';
        when (6) modus='JPEG';
        otherwise modus='unknown';
    end;
    comment=modus;
    compress=score;
end;
when (262) do;
    text='PhotometricInterpretation';
    select (score);
        when (0) modus='MinSampleValue=White';
        when (1) modus='MaxSampleValue=White';
        when (2) modus='RGB';
        when (3) modus='PaletteColor';
        when (4) modus='TransparencyMask';
        when (6) modus='YCbCrColorSpace';
        when (8) modus='1976CIEL*a*b ColorSystem';
        otherwise modus='unknown';
    end;
    comment=modus;
    photo=score;
end;
when (263) text='Tresholding';
when (264) text='CellWidth';
when (265) text='CellLength';
when (266) text='FillOrderTag';
when (269) text='DocumentName';
when (270) text='ImageDescription';

```

```

when (271) text='ScannerMake';
when (272) text='ScannerModel';
when (273) do;
    text='StripOffsets';
    nstrips=length;
    kstrips=score;
end;
when (274) text='Orientation';
when (277) text='SamplesPerPixel';
when (278) do;
    text='RowsPerStrip';
    nrows=score;
end;
when (279) do;
    text='StripByteCounts';
    lstrips=score;
end;
when (280) do;
    text='MinSampleValue';
    minval=score;
end;
when (281) do;
    text='MaxSampleValue';
    maxval=score;
end;
when (282) do;
    text='XResolution';
    input @(score+1) z pib4. n pib4.;
    xres=z/n;
    comment=left(put(xres,best12.3));
end;
when (283) do;
    text='YResolution';
    input @(score+1) z pib4. n pib4.;
    yres=z/n;
    comment=left(put(yres,best12.3));
end;
when (284) do;
    text='PlanarConfiguration';
    planar=score;
    select (score);
        when (1) modus='NoPlaneSplitting';
        when (2) modus='PlaneSplitting';
        otherwise modus='unknown';
    end;
    comment=modus;
end;
when (285) text='PageName';
when (286) text='XPosition';
when (287) text='YPosition';
when (288) text='FreeOffsets';
when (289) text='FreeByteCounts';
when (290) text='GrayResponseUnit';
when (291) text='GrayResponseCurve';
when (292) text='T4Options';
when (293) text='T6Options';
when (296) do;
    text='ResolutionUnit';
    resunit=score;
    select (score);
        when (1) unit='Aspect ';

```

```

        when (2) unit='Inch  ';
        when (3) unit='Cm   ';
        otherwise unit='unknown';
    end;
    comment=unit;
end;
when (297) text='PageNumber';
when (300) text='ColorResponseUnit';
when (301) text='TransferFunction';
when (305) text='Software'; /* Ersteller Software */
when (306) text='DateTime'; /* Erstellung am um */
when (315) text='Artist'; /* Macher Name */
when (316) text='HostComputer'; /* Macher Computer */
when (317) text='Predictor'; /* nur bei LWZ-Compression=5 */
when (318) text='WhitePoint';
when (319) text='PrimaryCromaticities';
when (320) text='ColorMap';
when (321) text='HalftoneHints';
when (322) text='TileWidth';
when (323) text='TileLength';
when (324) text='TileOffsets';
when (325) text='TileByteCount';
when (332) text='InkSet';
when (333) text='InkNames';
when (334) text='NumberOfInks';
when (336) text='DotRange';
when (337) text='TargetPrinter';
when (338) text='ExtrasSamples';
when (339) text='SampleFormat';
when (340) text='SMinSampleValue';
when (340) text='SMinSampleValue';
when (341) text='SMaxSampleValue';
when (342) text='TransferRange';
when (512) text='JPEGProc';
when (513) text='JPEGInterchangeFormat';
when (514) text='JPEGInterchangeFormatLength';
when (515) text='JPEGRestartInterval';
when (517) text='JPEGLossLessPredictors';
when (518) text='JPEGPointTransforms';
when (519) text='JPEGQTables';
when (520) text='JPEGDCTables';
when (521) text='JPEGACTables';
when (529) text='YCbCrCoefficient';
when (530) text='YCbCrSubSampling';
when (531) text='YCbCrPositioning';
when (532) text='ReferenceBlackWhite';
otherwise text='UnknownTag';
end;
select (type);
when ( 1) typ='Byte  '; /* 8 Bit-Byte */
when ( 2) typ='ASCII  '; /* 8 Bit-ASCII-Code */
when ( 3) typ='SHORT  '; /* 16 Bit unsigned Integer */
when ( 4) typ='LONG   '; /* 32 Bit unsigned Integer */
when ( 5) typ='RATIONAL  '; /* 2 LONGs: Fraction/Denominator */
when ( 6) typ='SByte  '; /* 8 Bit signed Integer */
when ( 7) typ='UNDEFINED'; /* 8 Bit Anything */
when ( 8) typ='SSHORT  '; /* 16 Bit signed Integer */
when ( 9) typ='SLONG   '; /* 32 Bit signed Integer */
when (10) typ='RATIONAL  '; /* 2 SLONGs: Fraction/Denominator */
when (11) typ='FLOAT   '; /* 4 Byte IEEE floating point */
when (12) typ='DOUBLED  '; /* 8 Byte IEEE floating point */

```

```

        otherwise typ='Unknown ' ;
    end;
    output tags;
end;
do istrip=1 to nstrips;
    input @(4*istrip+kstrips-3) stripa pib4.
           @(4*istrip+lstrips-3) stripl pib4.;
    stripa=stripa+1;
    stripe=stripa+stripl-1;
    output stripes;
end;
output scores;
goto a;
run;

```

WORK.SCORES -- Auszug									
X	Y		C	O	N				R
P	P		M	S					E
I	I		P	P	T	N			S
X	X	B	R	H	R	R	X	Y	U
E	E	I	E	O	I	O	R	R	N
L	L	T	S	T	P	W	E	E	I
S	S	S	S	O	S	S	S	S	T
465	320	446606	1	2	30	11	299	288	2

WORK.TAGS							
OBS	TEXT	COMMENT	TAG	TYPE	LENGTH	SCORE	TYP
1	NewSubfile		254	4	1	0	LONG
2	ImageWidth	Xpixels	256	4	1	465	LONG
3	ImageLength	Ypixels	257	4	1	320	LONG
4	BitsPerSample		258	3	3	446606	SHORT
5	Compression	Uncompressed	259	3	1	1	SHORT
6	PhotometricInterpretation	R G B	262	3	1	2	SHORT
7	ScannerModel		272	2	10	446612	ASCII
8	StripOffsets		273	4	30	446622	LONG
9	SamplesPerPixel		277	3	1	3	SHORT
10	RowsPerStrip		278	4	1	11	LONG
11	StripByteCounts		279	4	30	446742	LONG
12	XResolution	299	282	5	1	446862	RATIONAL
13	YResolution	288	283	5	1	446870	RATIONAL
14	ResolutionUnit	Inch	296	3	1	2	SHORT
15	Software		305	2	25	446878	ASCII
16	DateTime		306	2	26	446903	ASCII

## WORK.STRIPES

OBS	STRIPA	STRIPL	STRIPE	OBS	STRIPA	STRIPL	STRIPE
1	<b>9</b>	<b>15345</b>	15353	16	230184	15345	245528
2	15354	15345	30698	17	245529	15345	260873
3	30699	15345	46043	18	260874	15345	276218
4	46044	15345	61388	19	276219	15345	291563
5	61389	15345	76733	20	291564	15345	306908
6	76734	15345	92078	21	306909	15345	322253
7	92079	15345	107423	22	322254	15345	337598
8	107424	15345	122768	23	337599	15345	352943
9	122769	15345	138113	24	352944	15345	368288
10	138114	15345	153458	25	368289	15345	383633
11	153459	15345	168803	26	383634	15345	398978
12	168804	15345	184148	27	398979	15345	414323
13	184149	15345	199493	28	414324	15345	429668
14	199494	15345	214838	29	429669	15345	445013
15	214839	15345	230183	30	445014	<b>1395</b>	<b>446408</b>

/\* DATA STEP zur Erzeugung der SAS-Datei (.RGB) \*/

```

data f26lmsk.rgb (keep=x y r g b); /* PhotometricInterpretation=RGB */
  length default=3;
  infile tif recfm=n;
  set scores;
  y=ypixels+1;
  do i=1 to nstrips;
    set stripes;
    k=stripa-3;
    do j=1 to nrows;
      do y=y-1 while(y>0);
        do x=1 to xpixels;
          k+3;
          input @k R pib1. G pib1. B pib1.;
          if R+G+B then output;
        end;
      end;
    end;
  end;
run;

```

F26LMSK.RGB -- Auszug					
OBS	Y	X	R	G	B
87783	76	379	128	107	117
87784	76	380	121	103	108
87785	76	381	125	112	114
87786	76	382	140	133	137
87787	75	64	148	139	154
87788	75	65	145	139	149



*/\* Erzeugung einer SEED-Datei für die anschließende Clusteranalyse \*/*

```
libname scholz "d:\mp120193";
```

```
DATA SCHOLZ.SEED;
  INPUT R G B CLUSTER;
  CARDS;
  165 150 155 1
  155 135 140 2
  145 125 135 3
  135 115 120 4
  125 110 115 5
  115 100 105 6
  105 95 100 7;
RUN;
```

*/\* Proedur Clusteranalyse \*/*

```
PROC FASTCLUS DATA =f26lmsk.RGB (WHERE=(R+G+B))
  SEED =SCHOLZ.SEED
  MAXC =7
  MAXITER =99
  DRIFT
  REPLACE =PART
  CONVERGE=0.001
  OUT =f26lmsk.CLUST;
  VAR R G B;
RUN;
```

F26LMSK.CLUST -- Auszug							
	↓	↓				↓	
OBS	Y	X	R	G	B	CLUSTER	DISTANCE
102523	3	303	130	107	115	6	4.8076
102524	3	304	132	110	117	5	6.4517
102525	3	305	130	110	117	5	7.4931
102526	3	306	132	114	118	4	3.8475
102527	3	307	139	122	126	4	3.2864
102528	3	308	142	124	133	4	4.7908
102529	3	310	137	120	128	4	4.6087
102530	3	311	134	116	123	5	2.7512
102531	3	312	133	113	115	5	6.1770
102532	3	313	129	110	111	6	4.8802
102533	3	314	143	126	130	4	4.4330
102534	3	315	148	132	135	3	2.5921

*/\* Proedur Diskriminanzanalyse \*/*

```
PROC DISCRIM DATA =f26lmsk.CLUST
  METHOD=NORMAL
  POOL =YES
  OUT =f26lmsk.DISKRI;
  CLASS CLUSTER;
  VAR R G B;
RUN ;
```

F26LMSK.DISKRI -- AUSZUG											
OBS	↓ Y	↓ X	R	G	B	↓ CLUSTER	DISTANCE	_1	...	_7	↓ _INTO_
102307	11	308	155	151	153	2	14.1864	0.54274	0.000000	0.000000	1
102308	11	309	147	136	146	2	7.3491	0.00000	0.000000	0.000000	2
102309	11	310	137	118	130	4	5.9974	0.00000	0.000000	0.000000	4
102310	11	311	131	111	119	5	5.3652	0.00000	0.000000	0.000000	5
102311	11	312	129	107	114	6	3.5556	0.00000	0.00000	0.0000076	6

*/\* ERZEUGUNG des Clusterbildes \*/*

```
libname scholz "d:\mp120193";
title "create the ANNOTATE - scholz02s.sas";


data anno;
  attrib x y length=4;
  length default=4;
  array farbe (0:15) $8 _temporary_ ('gray' 'white' 'blue' 'green'
    'cyan' 'yellow' 'magenta' 'orange'
    'red' 'brown' 'lime' 'gold'
    'tan' 'purple' 'pink' 'steel');
  retain function 'symbol';
  retain text 'dot';
  retain xsys ysys '2';
  retain size .023;
  set f26lmsk.DISKRI (where=( _into_ ));
  color=farbe(mod(_into_,16));
  keep function color text xsys ysys size x y;
run;
```

WORK.ANNO -- Auszug									
OBS	X	Y	FUNCTION	TEXT	XSYS	YSYS	SIZE	COLOR	
87501	97	76	symbol	dot	2	2	0.023000	white	
87502	98	76	symbol	dot	2	2	0.023000	blue	
87503	99	76	symbol	dot	2	2	0.023000	red	
87504	100	76	symbol	dot	2	2	0.023000	green	
87505	101	76	symbol	dot	2	2	0.023000	green	
87506	102	76	symbol	dot	2	2	0.023000	blue	

```
goptions reset=all dev=win;
goptions colors=('gray' 'white' 'blue' 'green'
  'cyan' 'yellow' 'magenta' 'orange'
  'red' 'brown' 'lime' 'gold'
  'tan' 'purple' 'pink' 'steel') cback=black;
proc ganno anno=_last_ datasys;
run;
```

## 7-Cluster-Bild

weiß charakterisiert intramuskuläres Fett




### Ergebnisse aus der Cluster- bzw. Diskriminanzanalyse

#### Cluster Means

Cluster	R	G	B
1	163.980	155.374	163.413
2	153.553	138.978	145.278
3	146.959	129.780	136.092
4	140.810	122.177	128.498
5	134.524	114.412	120.608
6	127.470	105.206	111.080
7	118.608	92.853	97.597

#### Cluster Standard Deviations

Cluster	R	G	B
1	6.00447	8.18150	8.94079
2	3.53037	3.86297	4.31256
3	3.05880	3.00326	3.33203
4	2.95175	2.84208	3.23728
5	3.05158	3.03086	3.45572
6	3.33078	3.56051	4.08508
7	4.45324	5.25349	5.79337



102534 Observations	102533 DF Total
3 Variables	102527 DF Within Classes
7 Classes	6 DF Between Classes
Class Level Information	
Output	
CLUSTER	Frequency Proportion
1	2685 0.026186 <b>2.62 % IMF</b>
2	13450 0.131176
3	23744 0.231572
4	24667 0.240574
5	19679 0.191927
6	12702 0.123881
7	5607 0.054684

Das Cluster mit den höchsten Mittelwerten für die Farbkanäle R, G und B entspricht dem sichtbaren intramuskulären Fett. Der prozentuale Anteil der Pixel in diesem Cluster gibt folglich den Marmorierungsgrad an.

Die beiden Werte (zwei Kotelettseiten) für den Marmorierungsgrad aus der Videobildanalyse wurden anschließend als potentielle Regressoren in einer Multiplen Stepwise-Regressionsanalyse zur Schätzung des intramuskulären Fettgehaltes verwendet.

Als Referenzwerte dienten die mit Hilfe der Nah-Infrarot-Transmissionsspektroskopie ermittelten Fettgehalte im Rückenmuskel. Diese Untersuchungen wurden mit einem INFRATEC-Gerät am Landesuntersuchungsinstitut für Lebensmittel, Arzneimittel und Tierseuchen (LAT) Berlin durchgeführt.

Ergebnisse des Crossvalidation-Testes (fitting sample, n=65 und validation sample, n=65)

Intramuskulärer Fettgehalt % aus NIT	
<i>fitting sample</i>	
<b>R<sup>2</sup></b>	<b>.706</b>
<b>SEE</b>	0,283
<i>validation sample</i>	
<b>SEP</b>	0,295

$R^2$  = Bestimmtheitsmaß; SEE = Standard Error of Estimation ( $\sqrt{MS \text{ residual}}$ ), SEP = standard error of prediction ( $\sqrt{(y_i - \hat{y}_i)^2 / n}$ ) - validation sample, DF = Freiheitsgrade - Anzahl der Variablen, die in der Regressionsgleichung verblieben

### Schlußfolgerungen

Die Videobildauswertung mit Hilfe der Cluster- bzw. Diskriminanzanalyse eröffnet die Möglichkeit, den Marmorierungsgrad im Kotelettmuskel beim Schwein oder anderen Tierarten relativ einfach, schnell und objektiv zu bestimmen. Voraussetzungen für eine erfolgreiche Verwendung dieser Technik sind die Einhaltung standardisierter Bedingungen bei der Bildaufnahme unter weitgehender Vermeidung von Oberflächenreflexionen. Es sollte eine möglichst hochauflösende Digitalkamera verwendet werden.

Literatur:

Referenzhandbuch Dateiformate

Günter Born

Addison Wesley, 4. Aufl. 1996, 557 - 592

Johnson, B. and W. Johnson (1992): Importing X-Window  
System Raster Images into SAS-Graph / Software Output.  
SUGI 17, Hawaii, 543 - 548

---

\* Herzlicher Dank gebührt Herrn W. Lesener (Rechenzentrum, HU-Berlin)